Hedonic Demand for Rented House in Kohima, Nagaland

SK Mishra*
ML Ngullie**

1. Introduction: Unlike his predecessor economists who in promulgating their own theories of consumer's behavior considered the demand for (or supply of) a commodity merely as an expressed willingness to participate in exchange activity of a quantum of any particular commodity for a quantum of some other commodity (or money), Lancaster (1966) considered a commodity as a bundle of characteristics or a bunch of vectors, $x = [x_1, x_2, ..., x_m]$. A particular commodity, $x^{(0)} = [x_1^{(0)}, x_2^{(0)}, ..., x_m^{(0)}]$ is different from another commodity $x^{(1)} = [x_1^{(1)}, x_2^{(1)}, ..., x_m^{(1)}]$ if and only if $x_j^{(0)} \neq x_j^{(1)}$ for at least one $j; j = 1, 2, ..., m.$ The demand for (or supply of) a commodity is therefore a demand for (or supply of) characteristics. These characteristics may include time (whether a commodity is old or new), place (its location), positional value (whether it is owned or used by many or only a few), brand name (whether produced by this or that manufacturer), and so on. This view of considering a commodity as a bundle of characteristics opens an immensely wide scope for properly dealing with the demand for (or supply of) a commodity not only as a substitute of but also as a complement to other commodity or commodities.

Rentable house, for example, is a commodity which has a demand (on rent) and often this demand is dependent on the house rent (per month, say), disposable income of the person (family) and the number of members in the family. Had all houses been exactly identical (in matters of location, number of rooms, number of floors, carpet area, available facilities, neighborhood characteristics, and so on) rent, income and family size (and such variables) would certainly have been sufficient to determine the demand for houses. But on the contrary each house differs from another house in at least one characteristic. Even if a house, $H^{(1)}$ is exactly the same as another house, $H^{(2)}$ in all characteristics, $j = 1, 2, ..., m-1,$ the m^{th} characteristic of it (namely location in the 3-d space) must always differ since it is impossible for two houses to occupy exactly the same location. Consumers may have (and often do have) strong preferences for location. This is an awkward situation for the traditional theory of consumer's demand although the characteristic theory can comfortably handle it.

2. The Objective and the Data Base: The objective of this study is to show how the demand for a rentable house can be quantitatively expressed. The data set is obtained by a primary (sample) survey of 209 households randomly selected from the households inhabiting 19 wards of the township of Kohima, the capital city of Nagaland (India) during the first half of the year 2008. Eleven households were chosen from each ward. Besides many other information, the survey collected data on the residential house of the respondent, his/her family size, family income and rent (per month) paid if the house was acquired on rent. Of 209 households, 109 were found living in a rented house. In this study we use the data for those 109 households.

Among the house characteristics, information on the following were collected: (1) House type – kutcha, pucca-a or pucca-b, (2) plot size – sq. ft, (3) floor area – sq. ft, (4) no. of rooms, (5) no. of occupants – persons, (6) nature of ownership – rented, govt. quarter, own, (7) distance from the nearest building – ft, (8) receiving enough sun shine – no, yes, (9) parking space – no, yes, (10) waste disposal facilities – no, yes-near, yes-far, (11) drainage – no, ordinary, very good, (12) public garden/park nearby – no, yes, (13) having water supply – no, yes, (14) regularity of water supply – not satisfactory, satisfactory, very good, (15) source of water supply – outside common, outside private, inside, (16) nature of toilet – outside common, outside private, inside attached, (17) power connection – no yes, (18) load-shedding or power failure – frequent, occasionally, rarely, (19) noise pollution – no, yes, (20) air pollution – no, yes, (21) water pollution – no, yes, (22) nature of water pollution – physical, chemical, both, (23) respondent's feeling of satisfaction with the house – no, yes, and (24) safety – unsatisfactory, satisfactory . Most of

---

* Department of Economics, North-Eastern Hill University, Shillong (India)
** Department of Economics, Dimapur Govt. College, Dimapur, Nagaland (India)

these characteristics are qualitative in nature and we have used binary variables or an ordinal/nominal scale to measure them such that higher value on the scale refers to a preferred state and vice versa. Further, of these characteristics, we have dropped #6 since in the subsequent analysis we are concerned with rented house only. In our analysis, rent is used as a price (in Rs.) per sq. ft. of floor area.

3. Methodological Aspects: We hold that three variables, namely, household (disposable) income (Y), family size (F), and the monthly rental (R) should explain the demand for housing. We also hold a-priori that the coefficient/exponent associated with rent should be negative while the coefficients associated with income, and family size, should be positive – the first measuring the ability to pay and the second measuring a need for larger house which may require higher rent to be paid.

The crudest of the possibilities is to regress each of the characteristics of houses on Y, F and R. In that case we will have 22 regression equations to be estimated which will involve 88 parameters (including 22 intercepts). It will be difficult to cogently explain those parameters. Further, this approach may not be suitable in view of interdependencies (substitution as well complementation) among various characteristics since we must consider each equation independently. Simultaneous equation models (such as seemingly unrelated regression equations method of estimation) may not be applicable in want of identifiability and unknown nature of residuals.

Another way to establish the relationship between the demand for house and its determinants is to use factor analysis/principal components analysis to identify the leading factors in the complex of all house characteristics, regress the factor scores on the determinants (Y, F and R) and back-calculate the coefficients from the factor weights matrix and the regression coefficients. Alternatively, it is also possible that canonical correlations between housing characteristics (X) and the determinants, $Z=(Y,F,R)$, of house demand are obtained and canonical factor scores are used to back-calculate the coefficients of the demand equation(s). These methods heavily rely on the correlation matrix, $\Re(X), \Re(Z)$ or both. The degree of success in obtaining the demand equation(s) will depend on the eigen-structure these matrices.

Yet another method was suggested by Stackelberg (1932), although in a different context (Baumgärtner, 2001; Barrett and Hogset, 2003). In the Stackelberg schema, estimation of the demand function for a multi-characteristics commodity is rather simple. Let $X_i=(x_{i1}, x_{i2}, ..., x_{im});\ i=1,2,...,n$ be n observations on m joint housing characteristics. Then, a point r_i in m-dimensional (real) space is defined as $r_i=(x_{i1}^2+x_{i2}^2+...+x_{im}^2)^{0.5}$. Only the positive value is taken (since r_i signifies length). One of the characteristics (say, x_1) is considered as a standard measure or reference. The direction vectors for other characteristics are obtained as $\theta_{ij}=\cos^{-1}(x_{ij}/r_i);\ j=2,3,...,m.$ These direction vectors together with the determinants of housing demand $[Z=(Y,F,R)]$ are used as regressors and r is used as the regressand variable. Thus, $r_i=f(\theta_{i2},\theta_{i3},...,\theta_{im};Y_i,F_i,R_i);\ i=1,2,...,n$ are used to estimate the parameters of demand function of a multi-characteristics commodity such as house. Note that One of the characteristics, here the first one, is used as a standard, and hence θ_1 is not included among the regressors. Implicitly it is assumed that all the characteristics are identically related to the determinants, $Z=[Y,F,R]$. It is also assumed that all the products are measured in the same (Euclidean) space. Since different characteristics are measured in different scales/units, it is appropriate to normalize each of them to have unit norm. We define $norm_j=\left[\sum_{i=1}^{n}x_{ij}^2\right]^{0.5}$ and $x_{ij}=x_{ij}^*/norm_j; i=1,n; j=1,m,$ where x^* is measured in variant units and x is measured with the unit norm. We use this method in our study.

4. Results and Discussion: In our analysis, we have used the floor area of house as the reference characteristics. We have used the direction vectors of all characteristics (sans no. of occupants and nature of ownership, since both of these characteristics are implicitly taken into consideration as we are analyzing the case of rented houses only and the family size of the household is used as an non-characteristic explanatory variable). All variables (dependent, r, as well as independent, θ and Z) are transformed into their (natural) logarithmic values. The results are presented in table-1. Among the house-characteristics variables (direction vectors) only 'house type' is statistically significant at 10% (one-tail) level of significance; others are significant al lower levels. Among the non-characteristics variables, family size is statistically significant at 10% (one-tailed) level. It may be noted that the source of water supply and power connection are complementing floor area and have positive coefficients; other characteristics are substitutes of the floor area as they bear negative coefficients. The negative coefficient associated with rent is not (statistically) distinguishable from zero.

Table-1. Log-Linear Regression Coefficients of Demand function for House in Kohima, Nagaland

Sl. No.	Variable	Coefficient	t-value	Sl. No.	Variable	Coefficient	t-value
1	**House type**	-0.1866	-1.42	14	**Nature of toilet**	-0.6106	-4.74
2	**Plot size**	-0.6201	-6.63	15	**Power connection**	1.9809	6.90
3	**Floor area (Ref Characteristics)**	-	-	16	**Load-shedding**	-0.3849	-2.55
4	**No. of rooms**	-0.6653	-4.10	17	**Noise pollution**	-0.2934	-4.42
5	**Distance from nearest building**	-0.6482	-7.29	18	**Air pollution**	-0.4547	-5.78
6	**Availability of sunlight**	-0.3846	-5.80	19	**Water pollution**	-0.4232	-2.33
7	**Parking space**	-0.5431	-7.64	20	**Nature of water pollution**	-0.6117	-3.28
8	**Waste disposal facilities**	-0.2366	-2.96	21	**Resident's satisfaction**	-0.3635	-4.60
9	**Drainage**	-0.4945	-5.01	22	**Safety feeling**	-0.2544	-3.04
10	**Adjacency of Park/public garden**	-0.5580	-8.85	23	**Income**	0.0270	2.29
11	**Water supply status**	-0.4963	-3.18	24	**Family size**	0.0174	1.39
12	**Regularity of water supply**	-0.7605	-6.98	25	**Rent per sq ft floor area**	-0.0009	-0.07
13	**Source of water supply**	0.4528	1.70	26	**Regression constant**	1.6642	3.81

R square=0.936787; Table values of 2-tailed t = 0.68 (50%), 1.29 (20%), 1.66 (10%), 1.99 (5%), 2.37 (2.5%), 2.64 (1%).
Regression coefficients are indeed elastcities since regressors and regressand variable are all in natural logarithms.

.

Table-2. Income and Family Size Elasticity of Consumption Expenditure in Kohima

Item of Expenditure	Elasticity: Income	Elasticity: Family Size	Elasticity: Total	Constant	Item of Expenditure	Elasticity: Income	Elasticity: Family Size	Elasticity: Total	Constant
Cereals	3.01E-01	4.54E-01	7.55E-01	4.35E+01	Travel	2.31E+00	-7.15E-01	1.59E+00	1.93E-06
	6.73E+00	8.79E+00	1.55E+01	1.08E+01		9.07E+00	2.43E+00	1.15E+01	6.62E+00
Vegetables	5.50E-01	2.59E-02	5.76E-01	6.34E+00	Education fees	9.04E-01	3.00E+00	3.91E+00	2.15E-03
	1.00E+01	4.07E-01	1.04E+01	4.30E+00		3.79E+00	1.09E+01	1.47E+01	3.29E+00
Non-veg items	6.26E-01	-5.63E-03	6.20E-01	3.09E+00	Cable TV fees	2.01E+00	-6.72E-01	1.34E+00	9.61E-06
	6.54E+00	5.09E-02	6.59E+00	1.51E+00		8.13E+00	2.35E+00	1.05E+01	5.98E+00
Sugar	2.99E-01	4.00E-01	6.99E-01	2.89E+00	Telephone bills	1.99E+00	-6.97E-01	1.29E+00	2.32E-05
	4.57E+00	5.28E+00	9.85E+00	2.08E+00		7.91E+00	2.40E+00	1.03E+01	5.43E+00
Tea leaf	6.15E-01	1.45E-01	7.60E-01	4.32E-01	Guest entertainment	2.30E+00	-2.04E-01	2.10E+00	3.35E-07
	8.16E+00	1.66E+00	9.83E+00	1.43E+00		7.85E+00	6.02E-01	8.45E+00	6.50E+00
Milk	7.14E-01	1.26E-01	8.40E-01	6.66E-01	Hobbies	1.95E+00	-3.74E-01	1.58E+00	1.03E-06
	1.04E+01	1.58E+00	1.20E+01	7.59E-01		6.02E+00	9.97E-01	7.01E+00	5.45E+00
Edible oil	5.33E-01	8.78E-02	6.21E-01	1.11E+00	House rent	**-3.69E-01**	**-3.71E-01**	**-7.40E-01**	**1.46E+03**
	7.44E+00	1.06E+00	8.50E+00	1.93E-01		**8.61E-01**	**7.47E-01**	**1.61E+00**	**2.18E+00**
Fruits	1.47E+00	-1.89E-01	1.28E+00	6.28E-04	Toiletries	8.34E-01	-1.00E-01	7.33E-01	2.02E-01
	7.26E+00	8.09E-01	8.07E+00	4.67E+00		7.74E+00	8.05E-01	8.54E+00	1.90E+00
Water supply	1.23E+00	-4.54E-01	7.80E-01	9.72E-04	Addictive items	1.29E+00	-7.49E-01	5.39E-01	1.32E-03
	4.12E+00	1.31E+00	5.43E+00	2.96E+00		3.44E+00	1.73E+00	5.16E+00	2.26E+00
Fuel	4.55E-01	-3.91E-02	4.16E-01	9.47E+00	Clothes/ shoes	9.53E-01	4.95E-02	1.00E+00	2.13E-01
	9.16E+00	6.81E-01	9.84E+00	5.79E+00		1.39E+01	6.23E-01	1.45E+01	2.88E+00
Electricity	4.44E-01	4.70E-02	4.91E-01	3.55E+00	Medical bills	1.36E+00	1.94E-01	1.55E+00	3.44E-04
	4.26E+00	3.89E-01	4.64E+00	1.55E+00		4.32E+00	5.32E-01	4.85E+00	3.24E+00
Newspapers	1.47E+00	-6.58E-02	1.40E+00	1.95E-04	Social obligations	2.35E+00	-3.14E-01	2.04E+00	4.29E-07
	5.92E+00	2.30E-01	6.15E+00	4.42E+00		7.99E+00	9.24E-01	8.91E+00	6.38E+00

Note: The first row under each item is the measure of elasticity while the 2nd row under each item gives computed t values of the estimated elasticity of expenditure. The results are based on data obtained from 209 sample households. Source: Ngullie & Mishra, 2008

An significant but negative rent-elasticity of demand for rented house may suggest that it is a sticky commodity. This is substantiated by the negative income elasticity of demand (consumption expenditure) for rented houses (Ngullie and Mishra, 2008) although insignificantly different from zero (see table-2).

5. Concluding Remarks: This paper draws on the theory of consumer's demand from Kelvin Lancaster who suggested that a commodity may be considered as a bundle of numerous characteristics and consumers are willing to pay for those characteristics. In this paper we have shown how the demand for a multi-characteristics commodity such as house can be estimated by a method suggested (long back) by Stackelberg that transforms the measures of various characteristics into polar coordinates, and how this method may be useful in identifying the complementary and substitutive characteristics of the commodity concerned. We have not gone in for identification of the demand equation. As for Kohima, an analysis of primary data collected from the households inhabiting 19 wards of the town suggests that consumers of rented house consider floor area, water supply and power supply complementary to each other and other characteristics of house as substitutes of the floor area. It has also been found that in Kohima a rented house is possibly an inferior or sticky commodity and its income elasticity for the overall sample is negative, although statistically insignificant.

References

- Barrett, C. and Hogset, H. (2003), "Estimating Multiple-Output Production Functions for the CLASSES Model" see at http://aem.cornell.edu/special_programs/AFSNRM/Basis/ Documents/ Memos/Multi-output ProductionFunctionEstimationforCLASSES.pdf
- Baumgärtner, S. (2001), "Heinrich von Stackelberg on Joint Production", *Euro. J. History of Economic Thought*, 8 (4): 509-525.
- Lancaster, K. J. (1966), " A new approach to consumer theory", *J. of Political Economy*, 74(2): 132-157.
- Ngullie, M.L. and Mishra, S.K. (2008), "Structural Relations Among the Components of Household Income and Expenditure in Kohima, Nagaland", Working Paper Series, *SSRN*: http://ssrn.com/abstract=1215322
- Stackelberg, H. V. (1932), *Grundlagen einer reinen Kostentheorie* (Foundations of a Pure Theory of Costs). Verlag von Julius Springer, Wien.

```
      PROGRAM HEDONIC ! (BY SK MISHRA, NEHU, SHILLONG, INDIA)
      PARAMETER (N=109,MMAX=24,M=22,MX=3,MXX=25)
      !N=NO. OF OBSERVATIONS/HOUSEHOLDS,MMAX=TOTAL NO.OF CHARACTERISTICS
      !MX=TOTAL NO. OF CHARACTERISTICS INCLUDED IN THE ANALYSIS
      !MX=TOTAL NO. OF NON-CHARACTRISTICS EXPLANATORY VARIABLES NAMELY
      !(INCOME, FAMILY SIZE, RENT). MXX=MX+M
C     -----------------------------------------------------------------
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION Y(N,M),R(N),X(N,MXX),XX(MXX,MXX),XR(MXX),A(MXX),SE(MXX)
      DIMENSION T(MXX),RNORM(M+MX),ER(N),Z(N,MMAX),INCL(M)
C     -----------------------------------------------------------------
      ! LIST OF VARIABLES TO BE INCLUDED IN ANALYSIS (VAR 5,6 EXCLUDED)
      DATA (INCL(J),J=1,22)/0,1,2,4,!3RD VARIABLE IS USED AS A REFERENCE
     &     7,8,9,10,11,12,13,14,15,16,17,18,19,20,21,22,23,24/
C     -----------------------------------------------------------------
      OPEN(7,FILE='HOUSE.TXT') ! INPUT FILE
      DO I=1,N
      READ(7,*)(Z(I,J),J=1,MMAX),(X(I,M+J),J=1,MX)
      WRITE(*,*)(Z(I,J),J=1,MMAX),(X(I,M+J),J=1,MX)
      ENDDO
      CLOSE(7)
      ! PICK UP M VARIABLES FROM THE LARGER SET OF MMAX VARIABLES
      DO I=1,N
      DO K=2,M
      Y(I,K)=Z(I,INCL(K))
      ENDDO
      Y(I,1)=Z(I,3) ! USED AS A REFERENCE CHARACTERISTIC
C      WRITE(*,*) X(I,MXX),X(I,MXX)/Z(I,3)
      X(I,MXX)=X(I,MXX)/Z(I,3)! RENT PAID PER SQ FT OF FLOOR AREA
      ENDDO
      ! NORMALIZE THE CHARACTERISTICS VARIABLES
      DO J=1,M
      SS=0.D0
      DO I=1,N
      SS=SS+Y(I,J)**2
      ENDDO
      RNORM(J)=DSQRT(SS)
      DO I=1,N
      Y(I,J)=Y(I,J)/RNORM(J)
      ENDDO
      ENDDO
      ! NORMALIZE THE NON-CHARACTERISTICS VARIABLES
      DO J=1,MX
      SS=0.D0
      DO I=1,N
      SS=SS+X(I,J+M)**2
      ENDDO
      RNORM(J+M)=DSQRT(SS)
      DO I=1,N
      X(I,J+M)=X(I,J+M)/RNORM(J+M)
      ENDDO
      ENDDO
      ! FIND R (THE LENGTH OF CHARACTERISTICS)
      DO I=1,N
      SS=0.D0
      DO J=1,M
      SS=SS+Y(I,J)**2
      ENDDO
      R(I)=DSQRT(SS)
      ENDDO
      ! FIND DIRECTION VECTORS
      DO I=1,N
      DO J=2,M
      JJ=J-1
      X(I,JJ)=DACOS(Y(I,J)/R(I))
      ENDDO
      DO J=1,MX
```

```fortran
      X(I,J+M-1)=X(I,J+M)
      ENDDO
      X(I,MXX)=1.D0 ! SET THE LAST VARIABLE=1 TO TAKE CARE OF INTERCEPT
      ENDDO
C     LOG TRANSFORMATION ------------------------------------
      DO I=1,N
      R(I)=DLOG(R(I))
      DO J=1,M-1
      X(I,J)=DLOG(X(I,J))
      ENDDO
      DO J=M,MXX-1
      X(I,J)=DLOG(X(I,J))
      ENDDO
      ENDDO
C     -------------------------------------------------------
      OPEN(8,FILE='HCOS.TXT') ! OUTPUT FILE
      DO I=1,N
      WRITE(8,1)I,(X(I,J),J=1,6)
      ENDDO
      DO I=1,N
      WRITE(8,1)I,(X(I,J),J=7,12)
      ENDDO
      DO I=1,N
      WRITE(8,1)I,(X(I,J),J=13,18)
      ENDDO
      DO I=1,N
      WRITE(8,1)I,(X(I,J),J=19,24)
      ENDDO
      DO I=1,N
      WRITE(8,1)I,R(I)
      ENDDO
      CLOSE(8)
    1 FORMAT(I4,6F12.5)
      ! FIND COVARIANCE MATRICES
      DO J=1,MXX
      XR(J)=0.D0
      DO JJ=1,MXX
      XX(J,JJ)=0.D0
      DO I=1,N
      XX(J,JJ)=XX(J,JJ)+X(I,J)*X(I,JJ)
      ENDDO
      ENDDO
      DO I=1,N
      XR(J)=XR(J)+X(I,J)*R(I)
      ENDDO
      ENDDO
      CALL MINV(XX,MXX,DET) ! INVERT THE X'X MATRIX
      WRITE(*,*)'DET = ',DET
      ! FIND REGRESSION COEFFICIENTS
      DO J=1,MXX
      A(J)=0.D0
      DO JJ=1,MXX
      A(J)=A(J)+XX(J,JJ)*XR(JJ)
      ENDDO
      ENDDO
      ! FIND RESIDUALS, STANDARD ERRORS, T VALUES AND R SQUARE
      SUME=0.D0
      DO I=1,N
      EXPR=0.D0
      DO J=1,MXX
      EXPR=EXPR+X(I,J)*A(J)
      ENDDO
      ER(I)=(R(I)-EXPR)
      SUME=SUME+ER(I)**2
      ENDDO
      DO J=1,MXX
      T(J)=A(J)/DSQRT(SUME/(N-MXX)*XX(J,J))
```

```
      ENDDO
      ! PRINT COEFFICIENTS AND T VALUES
      WRITE(*,*)'REGRESSION COEFFICIENTS AND COMPUTED T VALUES'
      DO J=1,MXX
      WRITE(*,*)A(J),T(J)
      ENDDO
      WRITE(*,*)'THE LAST ROW IS INTERCEPT AND ITS COMPUTED T VALUE'
      CALL CORR(R,ER,N,RR)
      WRITE(*,*)'---------------------------------- '
      WRITE(*,*) 'R SQUARE =',1.D0-RR**2
      END
      ! COMPUTES INVERTED MATRIX
      SUBROUTINE MINV(A,N,D)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION A(N,N)
      U=1.D0
      D=U
      DO I=1,N
      D=D*A(I,I)
      A(I,I)=U/A(I,I)
         DO J=1,N
         IF(I.NE.J) A(J,I)=A(J,I)*A(I,I)
         ENDDO
         DO J=1,N
         DO K=1,N
         IF(I.NE.J.AND.K.NE.I) A(J,K)=A(J,K)-A(J,I)*A(I,K)
         ENDDO
         ENDDO
         DO J=1,N
         IF(J.NE.I) A(I,J)= -A(I,J)*A(I,I)
         ENDDO
      ENDDO
      RETURN
      END
C     COMPUTES CORRELATION COEFFICIENT
      SUBROUTINE CORR(X,Y,N,R)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION X(N),Y(N)
      SX=0.D0
      SSX=0.D0
      SY=0.D0
      SSY=0.D0
      SXY=0.D0
      DO I=1,N
      SX=SX+X(I)
      SY=SY+Y(I)
      SSX=SSX+X(I)**2
      SSY=SSY+Y(I)**2
      SXY=SXY+X(I)*Y(I)
      ENDDO
      CVX=SXY/N-(SX/N)*(SY/N)
      SDX=DSQRT(SSX/N-(SX/N)**2)
      SDY=DSQRT(SSY/N-(SY/N)**2)
      R=CVX/(SDX*SDY)
      RETURN
      END
```